This lab explores the use of finite difference methods to approximate the solution $y(x)$ to linear boundary value problems of the form

$$
y^{\prime \prime}=p(x) y^{\prime}+q(x) y+r(x) \quad \text { for } \quad a \leq x \leq b
$$

where $y(a)=\alpha$ and $y(b)=\beta$. Under the assumptions that $p, q$ and $r$ are continuous and moreover that $q>0$, it is theoretically known there is a unique solution $y(x)$ to the differential equation. It therefore makes sense to try and approximate that solution numerically.

Begin by recalling that the derivative operator $D$ satisfies

$$
D=\frac{1}{h} \log \mathcal{E}=\frac{2}{h} \log \left(\frac{1}{2}\left(\Delta_{0}+\sqrt{\Delta_{0}^{2}+4 I}\right)\right)
$$

where $\mathcal{E}$ and $\Delta_{0}$ are respectively the shift and central difference operators

$$
\mathcal{E} z_{k}=z_{k+1} \quad \text { and } \quad \Delta_{0} z_{k}=z_{k+\frac{1}{2}}-z_{k-\frac{1}{2}} .
$$

Since $\Delta_{0}=\mathcal{O}(h)$ and assuming $h$ is sufficiently small, a power-series expansion of log yields

$$
D^{2}=\frac{4}{h^{2}} \sum_{j=0}^{\infty} \sum_{\ell=0}^{\infty} \frac{1}{2 j+1}\binom{-1 / 2}{j} \frac{1}{2 \ell+1}\binom{-1 / 2}{j}\left(\frac{1}{2} \Delta_{0}\right)^{2 j+2 \ell+2} .
$$

Truncating the series to the leading term yields

$$
D^{2}=\frac{4}{h^{2}}\left(\frac{1}{2} \Delta_{0}\right)^{2}=\frac{1}{h^{2}} \Delta_{0}^{2}+\mathcal{O}\left(h^{2}\right) .
$$

This approximation will represent $y^{\prime \prime}$.
For the first derivative recall the formula

$$
D=\frac{1}{h} \Delta_{0} \Upsilon_{0} \sum_{j=0}^{\infty} \sum_{\ell=0}^{\infty}\binom{-1 / 2}{j} \frac{1}{2 \ell+1}\binom{-1 / 2}{\ell}\left(\frac{1}{2} \Delta_{0}\right)^{2 j+2 \ell}
$$

where $\Upsilon_{0}$ is the averaging operator

$$
\Upsilon_{0} z_{k}=\frac{1}{2}\left(z_{k+\frac{1}{2}}+z_{k-\frac{1}{2}}\right)
$$

Again truncate the series to one term to obtain

$$
D=\frac{1}{h} \Delta_{0} \Upsilon_{0}+\mathcal{O}\left(h^{2}\right)
$$

This shall be used to approximate $y^{\prime}$.

## Discretizing the Differential Equation

To discretize the differential equation divide the domain $[a, b]$ into $m+1$ equal pieces of size $h=(b-a) /(m+1)$. Consider the grid points $x_{k}=a+h k$ for $k=1, \ldots, m$. Let $y_{k}$ be approximation of the exact solution $y\left(x_{k}\right)$ at each grid point. Note the boundary conditions imply $y_{0}=\alpha$ and $y_{m+1}=\beta$.

Having defined $y_{k}$, now employ the approximations

$$
y^{\prime \prime}\left(x_{k}\right) \approx \frac{1}{h^{2}} \Delta_{0}^{2} y_{k} \quad \text { and } \quad y^{\prime}\left(x_{k}\right) \approx \frac{1}{h} \Delta_{0} \Upsilon_{0} y_{k}
$$

to write the differential equation as the difference equation

$$
\frac{1}{h^{2}} \Delta_{0}^{2} y_{k}=p\left(x_{k}\right) \frac{1}{h} \Delta_{0} \Upsilon_{0} y_{k}+q\left(x_{k}\right) y_{k}+r\left(x_{k}\right)
$$

for $k=1,2, \ldots, m$. Since

$$
\frac{1}{h^{2}} \Delta_{0}^{2} y_{k}=\frac{y_{k+1}-2 y_{k}+y_{k-1}}{h^{2}} \quad \text { and } \quad \frac{1}{h} \Delta_{0} \Upsilon_{0} y_{k}=\frac{y_{k+1}-y_{k-1}}{2 h}
$$

then writing $p_{k}=p\left(x_{k}\right), q_{k}=q\left(x_{k}\right)$ and $r_{k}=r\left(x_{k}\right)$ yields

$$
\frac{y_{k+1}-2 y_{k}+y_{k-1}}{h^{2}}=p_{k} \frac{y_{k+1}-y_{k-1}}{2 h}+q_{k} y_{k}+r_{k} .
$$

Equivalently,

$$
-y_{k+1}+2 y_{k}-y_{k-1}+\frac{h}{2} p_{k}\left(y_{k+1}-y_{k-1}\right)+h^{2} q_{k} y_{k}=-h^{2} r_{k}
$$

for $k=1,2, \ldots, m$.
The equations when $k=1$ and $k=m$ can be rewritten involving the boundary conditions $y_{0}=\alpha$ and $y_{m+1}=\beta$ as

$$
-y_{2}+2 y_{1}+\frac{h}{2} p_{1} y_{2}+h^{2} q_{1} y_{1}=\alpha+\frac{h}{2} p_{1} \alpha-h^{2} r_{1} .
$$

and

$$
2 y_{m}-y_{m-1}-\frac{h}{2} p_{m} y_{m-1}+h^{2} q_{m} y_{m}=\beta-\frac{h}{2} p_{m} \beta-h^{2} r_{m} .
$$

The terms with $y_{k}$ have been written on the left and the boundary terms placed on the right. The result is a system of $m$ linear equations in the $m$ unknowns given by $y_{k}$ for $k=1,2, \ldots, m$.

To solve for the $y_{k}$, write the system in matrix form as $A y=c$ where

$$
A=\left[\begin{array}{ccccc}
2+h^{2} q_{1} & -1+\frac{h}{2} p_{1} & 0 & \cdots & 0 \\
-1-\frac{h}{2} p_{2} & 2+h^{2} q_{2} & -1+\frac{h}{2} p_{2} & 0 & \vdots \\
0 & \ddots & \ddots & \ddots & 0 \\
\vdots & 0 & -1-\frac{h}{2} p_{m-1} & 2+h^{2} q_{m-1} & -1+\frac{h}{2} p_{m-1} \\
0 & \cdots & 0 & -1-\frac{h}{2} p_{m} & 2+h^{2} q_{m}
\end{array}\right]
$$

and

$$
c=\left(\alpha+\frac{h}{2} p_{1} \alpha-h^{2} r_{1},-h^{2} r_{2}, \ldots,-h^{2} r_{m-1}, \beta-\frac{h}{2} p_{m} \beta-h^{2} r_{m}\right) .
$$

## Sparse Matrices

The matrix $A$ corresponding to the finite difference scheme has a lot of zeros. In particular only the diagonal, subdiagonal and supradiagonal entries contain coefficients of the system. For an $m \times m$ matrix this means that $3 m-2$ entries out of $m^{2}$ are non-zero. For example, if $m=128$ then

$$
\frac{3 m-2}{m^{2}}=\frac{382}{16384} \approx 2.3 \text { percent }
$$

of the entries are non-zero. This is called a sparse matrix.
As it would be wasteful to store all those zeros in memory and even more wasteful to compute with them, Julia includes a library for working such matrices called SparseArrays.

One way create a sparse matrix in Julia is to first create a regular matrix A and then convert it to a sparse matrix by removing the zeros with $A=s p a r s e(A)$. This technique is demonstrated as follows:

[^0]```
julia> A=diagm([1.0,2,3,4])
\(4 \times 4\) Matrix\{Float64\}:
    \(\begin{array}{llll}1.0 & 0.0 & 0.0 & 0.0\end{array}\)
    \(\begin{array}{llll}0.0 & 2.0 & 0.0 & 0.0\end{array}\)
    \(\begin{array}{llll}0.0 & 0.0 & 3.0 & 0.0\end{array}\)
    \(\begin{array}{llll}0.0 & 0.0 & 0.0 & 4.0\end{array}\)
julia> A=sparse(A)
\(4 \times 4\) SparseMatrixCSC\{Float64, Int64\} with 4 stored entries:
    \(\begin{array}{ccc}1.0 & . & . \\ . & 2.0 & .\end{array}\)
                        3.0 .
                        4.0
```

There is an obvious drawback to this technique because the initial step of constructing the matrix could take too much memory. A better way constructs the sparse matrix directly by specifying only the non-zero entries. This can by done with sparse(xs,ys,axy) which creates a matrix with entries $a_{i j}$ such that

$$
a_{i j}= \begin{cases}\operatorname{axy}[\mathrm{k}] & \text { for } i=\mathrm{xs}[\mathrm{k}] \text { and } j=\mathrm{ys}[\mathrm{k}] \\ 0 & \text { otherwise. }\end{cases}
$$

For example,

```
julia> A=sparse(1:4,1:4,[1.0,2,3,4])
4\times4 SparseMatrixCSC{Float64, Int64} with 4 stored entries:
\begin{tabular}{cccc}
1.0 & \(\cdot\) & \(\cdot\) & \(\cdot\) \\
\(\cdot\) & 2.0 & \(\cdot\) & \(\cdot\) \\
\(\cdot\) & \(\cdot\) & 3.0 & \(\cdot\) \\
\(\cdot\) & \(\cdot\) & \(\cdot\) & 4.0
\end{tabular}
```

constructs the same sparse matrix but without the intermediate step that requires allocating the memory needed for a full $m \times m$ array.

Sometimes, specifying lists of indices and values is tricky. A compromise solution is to first declare a sparse matrix with no non-zero entries and then fill in the needed values using an assignment within a loop.

```
julia> A=spzeros(4,4)
4\times4 SparseMatrixCSC{Float64, Int64} with 0 stored entries:
julia> for i=1:4
    A[i,i]=i
    end
julia> A
4\times4 SparseMatrixCSC{Float64, Int64} with 4 stored entries:
\begin{tabular}{cccc}
1.0 & \(\cdot\) & \(\cdot\) & \(\cdot\) \\
\(\cdot\) & 2.0 & \(\cdot\) & \(\cdot\) \\
. & \(\cdot\) & 3.0 & \(\cdot\) \\
. &. & \(\cdot\) & 4.0
\end{tabular}
```


## Approximating a Solution

We are now ready to approximate a solution to a linear two-point boundaryvalue problem using finite differences.

Each person will solve a different boundary value problem. In particular, your individualized problem will consist of different values $a, b, \alpha$, and $\beta$ for the boundary conditions and different functions $p(x), r(x)$ and $q(x)$ for the differential equation. Click on the following link to retrieve your boundary value problem
https://fractal.math.unr.edu/~ejolson/467-23/ab/mkab.cgi
Please do not use anyone else's differential equation for this lab.
The rest of this lab consists of a walk through demonstrating how to use the finite differences to approximate a solution of this differential equation. When I clicked on the link I obtained

```
Your differential equation is
    \mp@subsup{y}{}{\prime\prime}=p(x) y' + q(x) y +r(x) where a m x mb
with
    p(x)=-0.80
    q(x)=1.18+0.27*x^2
    r(x)=sin(1.43*x)
    a=0.92
    b=3.05
and boundary conditions
    y(a)=0.82
    y(b)=0.22
```

Create a subdirectory called lab05 and create the file finite. jl in that directory using gedit or some other program editor. Begin by defining the functions and parameters in the problem. Also load the LinearAlgebra and SparseArrays libraries as we will be using them later.

```
using LinearAlgebra,SparseArrays
p(x)=-0.80
q(x)=1.18+0.27*x^2
r(x)=sin(1.43*x)
a=0.92
b=3.05
alpha=0.82
beta=0.22
```

When performing numerical computations, it can be easy to lose track of what output corresponds to which input. Let's use the Symbolics library to print out the details of the differential equation before solving it.

```
@variables x
println("p(x)=",p(x))
println("q(x)=",q(x))
println("r(x)=",r(x))
println("y($a)=",alpha)
println("y($b)=",beta)
```

At this point it would be reasonable to test the program by opening a terminal window, changing to the lab05 subdirectory, starting Julia and then typing include("finite.jl"). The output should look similar to

```
julia> include("finite.jl")
p(x)=-0.8
q(x)=1.18 + 0.27(x^2)
r(x)=sin(1.43x)
y(0.92)=0.82
y(3.05)=0.22
```

Again, including the problem being solved as part of the output helps avoid errors. While there is not much room for getting the output of one program confused with another in a laboratory activity such as this one, such things are surprisingly important in practice.

Next, specify how many grid points will be used for the computation. Take $m=32$ which is hopefully large enough. For extra credit you may perform a convergence study for your problem by repeating the calculation for different values of $m$ and checking how fast the solution converges as $m$ increases. What is the observed order of convergence?

We now create the matrix $A$ and vector $c$ needed for the equation $A y=c$ to solve for $y$. Since $c$ is easier to construct first do that.

```
m=32
h=(b-a)/(m+1)
x=a.+(1:m)*h
c=-h^2*r.(x)
c[1]+=alpha+h/2*p(x[1])*alpha
c[m]+=beta-h/2*p(x[m])*beta
```

Note that c is built in stages. Line 24 initializes the entire vector with the term $-h^{2} r_{k}$. Then line 25 adds the boundary term $\alpha+\frac{h}{2} p_{1} \alpha$ to the first entry and line 26 adds $\beta-\frac{h}{2} p_{m} \beta$ to the last.

The matrix $A$ may be created in a similar way by initializing the diagonal and then adding supradiagonal and subdiagonal with loops.

```
A=sparse(1:m,1:m,2.0.+h^2*q.(x))
for i=1:m-1
    A[i,i+1]=-1+h/2*p(x[i])
end
for i=2:m
    A[i,i-1]=-1-h/2*p(x[i])
end
```

Finding the approximation $y_{k}$ for $k=1,2, \ldots, m$ can now be performed with the command $\mathrm{y}=\mathrm{A} \backslash \mathrm{c}$ which uses the built-in matrix libraries of Julia to efficiently solve the sparse linear algebra problem.

To finish this lab please print out the value of $y_{16}$ and draw a graph of the final approximation for your individualized differential equation. Code to do this for the example problem follows:
$\mathrm{y}=\mathrm{A} \backslash \mathrm{c}$
println("y(",x[16],")=",y[16])
using Plots
plot( $x, y$, label="m=\$m")
scatter!([a,b],[alpha,beta],label="boundary")

Once everything works, use the command savefig("graph05.pdf") to save the graph. It should look similar to


If your graph looks exactly like the above figure, that may mean you forgot to change the differential equation to the individual one downloaded from the link mentioned earlier.

## Submitting Your Work

Two things should be uploaded for grading:

- A PDF file lorenz.pdf containing the code finite.jl and the output from running that code.
- The graph graph05. pdf corresponding to your boundary value problem.

The file graph05.pdf can be created by adding savefig("graph05.pdf") to the end of your program. The only thing left is to convert lorenz.jl and its output into a PDF file for upload. In the lab the commands

```
$ julia finite.jl >finite.out
$ j2pdf -o finite.pdf finite.jl finite.out
```

may be used to produce a file finite.pdf suitable for uploading. You may check your submission using evince to view the PDF files.

Before leaving don't forget to close the applications open on your desktop and logout. Exit the Julia Repl by typing 〈ctrl〉-d and then $\langle c t r l\rangle-\mathrm{d}$ again to close the terminal. The editor has a menu at the top. If using one of the lab computers, please reboot it into Microsoft Windows.


[^0]:    julia> using LinearAlgebra,SparseArrays

